
/*
  Light Sensor Controls Servo and LED Strip; system turns on-off with Grove Switch.

  This code integrates a Grove light sensor (photocell) and a Grove switch to control
  a servo motor and an Adafruit NeoPixel LED strip. The servo motor oscillates between
  0 and 180 degrees, and the LED strip changes color based on the light level detected
  by the light sensor.

  A Grove switch is used to control the entire system:
  - When the switch is set to HIGH, the system operates as described below.
  - When the switch is set to LOW, the servo motor stops at 90 degrees, and the LED strip turns off.

  Hardware Setup:
  - Grove light sensor connected to analog pin A0
  - Grove switch connected to digital pin D3
  - Servo motor connected to pin D7
  - Adafruit NeoPixel LED strip connected to pin D5

  The system operates as follows:
  - When the light sensor detects a light level above a threshold (adjustable with the value 100), and the switch is set to HIGH, the servo motor oscillates between 0 and 180 degrees to create a sweeping motion. The LED strip lights up in green, providing visual feedback.
  - When the light level falls below the threshold, and the switch is set to HIGH, the servo motor remains still at 90 degrees, and the LED strip turns off in blue.
  - When the switch is set to LOW, regardless of the light level, the servo motor stops at 90 degrees, and the LED strip turns off.

*/

#include <Servo.h>
#include <Adafruit_NeoPixel.h>

#define LIGHT_SENSOR_PIN A0  // Use analog pin A0 for the light sensor
#define SWITCH_PIN 3        // Use digital pin D3 for the Grove switch

#define SERVO_PIN 7
Servo myservo;

#define LED_PIN 5           // Use the physical pin number for D5
#define NUM_LEDS 15         // Number of LEDs in the strip

Adafruit_NeoPixel strip = Adafruit_NeoPixel(NUM_LEDS, LED_PIN, NEO_GRB + NEO_KHZ800);

void setup() {
  Serial.begin(9600);
  myservo.attach(SERVO_PIN);
  pinMode(SWITCH_PIN, INPUT_PULLUP);  // Set the switch pin as an input with internal pull-up resistor
  strip.begin();
  strip.show();  // Initialize all pixels to 'off'
}

void loop() {
  // Read the state of the switch
  int switchState = digitalRead(SWITCH_PIN);

  if (switchState == HIGH) {
    int lightValue = analogRead(LIGHT_SENSOR_PIN);
    Serial.print("Light Sensor Value: ");
    Serial.println(lightValue);

    if (lightValue > 100) {
      for (int angle = 0; angle <= 180; angle += 5) {
        myservo.write(angle);
        delay(50);
      }
      for (int angle = 180; angle >= 0; angle -= 5) {
        myservo.write(angle);
        delay(50);
      }
      
      // Set LED strip to green
      strip.fill(strip.Color(0, 255, 0));  // Green color
      strip.show();
    } else {
      myservo.write(90);
      
      // Set LED strip to blue
      strip.fill(strip.Color(0, 0, 255));  // Blue color
      strip.show();
    }
  } else {
    myservo.write(90);
    
    // Turn off LED strip (blue)
    strip.fill(strip.Color(0, 0, 0));  // Turn off LEDs
    strip.show();
  }

  delay(500);
}




